

Mathematica code to solve a certain degree 10 Diophantine equation in 3 variables under some conditions - Ziqing Xiang

This is the Mathematica code for the joint work with Eiichi Bannai, Etsuko Bannai, Wei-Hsuan Yu and Yan Zhu entitled "Classification of spherical 2-distance {4, 2, 1}-design".

Section 3

Definition of Gegenbauer polynomial $Q_{n,4}(\xi)$.

```
In[1]:= Q[n_, 4, ξ_] := n (n + 6) / 24 ((n^2 + 6 n + 8) ξ^4 - 6 (n + 2) ξ^2 + 3) ;
```

Definition of F . Eq. (3.3).

```
In[2]:= F = Factor[Q[n, 4, 1] + k Q[n, 4, y/k] + (v - k - 1) Q[n, 4, (-y - 1)/(v - k - 1)] /.  
v -> 1/μ (k - x) (k - y) /. k -> μ - x y /. n -> (μ - x y) (μ - x y - x) (x + 1) / μ / (x - y) ]
```

```
Out[2]= ((x + x y - μ) (y + x y - μ)^2  
(x^2 y + x^3 y + x^2 y^2 + x^3 y^2 + 5 x μ - x^2 μ - 6 y μ - 2 x y μ - 2 x^2 y μ + μ^2 + x μ^2)  
(x^2 y + x^3 y + x^2 y^2 + x^3 y^2 + 3 x μ - x^2 μ - 4 y μ - 2 x y μ - 2 x^2 y μ + μ^2 + x μ^2)  
(x^5 y^2 + 2 x^6 y^2 + x^7 y^2 - x^2 y^3 - 2 x^3 y^3 - x^4 y^3 + x^5 y^3 + 2 x^6 y^3 + x^7 y^3 - x^2 y^4 - 2 x^3 y^4 -  
x^4 y^4 - 5 x^4 y μ - 7 x^5 y μ - 2 x^6 y μ - x y^2 μ - 2 x^2 y^2 μ + 3 x^3 y^2 μ - x^4 y^2 μ - 8 x^5 y^2 μ -  
3 x^6 y^2 μ + 2 y^3 μ + 4 x y^3 μ - x^2 y^3 μ - 3 x^3 y^3 μ + 6 x^3 μ^2 + 5 x^4 μ^2 + x^5 μ^2 - 2 x y μ^2 -  
7 x^2 y μ^2 + 6 x^3 y μ^2 + 10 x^4 y μ^2 + 3 x^5 y μ^2 + 3 y^2 μ^2 + 8 x y^2 μ^2 + 3 x^2 y^2 μ^2 - 3 x μ^3 -  
6 x^2 μ^3 - 4 x^3 μ^3 - x^4 μ^3 + y μ^3 + x y μ^3)) / (24 (1 + x)^2 (x - y)^4 (x y - μ)^2 μ^4)
```

The four factors F_0, F_1, F_2, F_3 of F . Eqs. (3.4) - (3.7).

```
In[3]:= F0 = (μ - x - x y) (μ - y - x y)^2 / (24 μ^4 (1 + x)^2 (x - y)^4 (μ - x y)^2) ;  
F1 = μ^2 + 5 μ x + μ^2 x - μ x^2 - 6 μ y - 2 μ x y + x^2 y - 2 μ x^2 y + x^3 y + x^2 y^2 + x^3 y^2 ;  
F2 = μ^2 + 3 μ x + μ^2 x - μ x^2 - 4 μ y - 2 μ x y + x^2 y - 2 μ x^2 y + x^3 y + x^2 y^2 + x^3 y^2 ;  
F3 = 3 μ^3 x + 6 μ^3 x^2 - 6 μ^2 x^3 + 4 μ^3 x^3 - 5 μ^2 x^4 + μ^3 x^4 - μ^2 x^5 - μ^3 y + 2 μ^2 x y - μ^3 x y + 7 μ^2 x^2 y -  
6 μ^2 x^3 y + 5 μ x^4 y - 10 μ^2 x^4 y + 7 μ x^5 y - 3 μ^2 x^5 y + 2 μ x^6 y - 3 μ^2 y^2 + μ x y^2 - 8 μ^2 x y^2 +  
2 μ x^2 y^2 - 3 μ^2 x^2 y^2 - 3 μ x^3 y^2 + μ x^4 y^2 - x^5 y^2 + 8 μ x^5 y^2 - 2 x^6 y^2 + 3 μ x^6 y^2 - x^7 y^2 - 2 μ y^3 -  
4 μ x y^3 + x^2 y^3 + μ x^2 y^3 + 2 x^3 y^3 + 3 μ x^3 y^3 + x^4 y^3 - x^5 y^3 - 2 x^6 y^3 - x^7 y^3 + x^2 y^4 + 2 x^3 y^4 + x^4 y^4 ;
```

Verify the factorization.

```
In[7]:= Simplify[F0 F1 F2 F3 - F]
```

```
Out[7]= 0
```

Section 5.2

Alternative definition of n and v . Eqs. (5.1) and (5.2).

```
In[8]:= n = (x + 1) (μ - x y) (μ - x y - x) / (μ (-y + x)) ;
v = n + 1 - (y n + μ - x y) / x ;
```

Step 1

Assumption on y in Step 1. Eq. (5.3).

```
In[10]:= yassum = y ≤ - (2 x^3 + 3 x^2 + 3 x + 2) || - (2 x^3 + 3 x^2 - 3 x - 3) ≤ y ≤ -1 ;
```

The computer proof for Step 1.

```
In[11]:= Simplify[v > n (n + 3) / 2 || F3 > 0, x ≥ 1 && μ ≥ 1 && yassum]
```

```
Out[11]= True
```

Step 2

Definition of a . Eq. (5.4).

```
In[12]:= usub = μ → - (x + a) y ;
```

Definition of G_1 .

```
In[13]:= G1 = Factor[(F3 /. usub) / y^2] ;
```

The computer proof for Step 2(a).

```
In[14]:= Simplify[(G1 /. y → 0) < 0, 2 ≤ x && -x < a]
```

```
Out[14]= True
```

The computer proof for Step 2(b).

```
In[15]:= Simplify[(G1 /. y → -1) > 0, 2 ≤ x && -x < a]
```

```
Out[15]= True
```

The computer proof for Step 2(c).

```
In[16]:= Simplify[(G1 /. y → - (2 x^3 + 3 x^2 + 3 x + 2)) > 0, 2 ≤ x && (-x < a ≤ -1 || 3 ≤ a)]
```

```
Out[16]= True
```

Step 3

Definition of b . Eq. (5.11).

```
In[17]:= ysub = y → - (2 x^3 + 3 x^2 + 3/2 (-1 + a) a x - 3/2 (-1 + a)^2 a + 1/4 x 3 (-1 + a) a (2 - 4 a + 3 a^2) -
1/4 x^2 3 (-1 + a) a^2 (3 - 6 a + 4 a^2) + 1/8 x^3 3 (-1 + a) a (5 - 9 a + 16 a^2 - 20 a^3 + 11 a^4) + b/x^4) ;
```

Definition of G_2 .

In[18]:= **G2 = Factor**[G1 /. ysub];

The computer proof for Step 3(a).

In[19]:= **Simplify**[(G2 /. b → -3994) > 0, 90 ≤ x && -1 ≤ a ≤ 3]

Out[19]= True

The computer proof for Step 3(b).

In[20]:= **Simplify**[(G2 /. b → 64) < 0, 90 ≤ x && -1 ≤ a ≤ 3]

Out[20]= True

Step 4

Definition of m^2 . Eq. (5.12).

In[21]:= **mmsub** = mm → n - (4 x^2 + 4 x - 2);

Definition of G_3 .

In[22]:= **G3 = Factor**[mm /. mmsub /. usub /. ysub];

Definition of \tilde{m}^2 . Eq. (5.12).

In[23]:= **mmtsub** = mmt → $a^2 - \frac{(-1+a)a^2}{x} + \frac{(-1+a)a(1+a^2)}{x^2} - \frac{1}{2x^3}(-1+a)a(1+2a+2a^3) + \frac{1}{4x^4}(-1+a)a(7-a+4a^2+4a^4) + \frac{c}{x^5};$

Definition of G_4 .

In[24]:= **G4 = Factor**[mmt /. mmtsub];

The computer proof for Step 4(a).

In[25]:= **Simplify**[G3 > (G4 /. c → -1620), 90 ≤ x && -1 ≤ a ≤ 3 && -3994 ≤ b ≤ 64]

Out[25]= True

The computer proof for Step 4(b).

In[26]:= **Simplify**[G3 < (G4 /. c → 3), 90 ≤ x && -1 ≤ a ≤ 3 && -3994 ≤ b ≤ 64]

Out[26]= True

The computer proof for Step 4(c).

In[27]:= **Simplify**[G4 < 9, 90 ≤ x && -1 ≤ a ≤ 3 && -1620 ≤ c ≤ 3]

Out[27]= True

Step 5

Definition of \tilde{n} . Eq. (5.16).

In[28]:= **nt** = mmt + (4 x^2 + 4 x - 2);

Definition of \tilde{v} . Eq. (5.16).

In[29]:= $vt = nt + 1 - (y \, nt + \mu - x \, y) / x;$

Definition of \tilde{z} . Eq. (5.15).

In[30]:= $zt = 144 \, mmt - (3 \, vt + (8 \, y + 4 \, x^3 + 6 \, x^2 + 3) (2 \, x + 1) - 3 / 2 \, mmt (mmt - 7)) ^2;$

Definition of G_5 .

In[31]:= $G5 = \text{Factor}[zt /. usub /. ysub /. mmtsub];$

The computer proof for Step 5(a). We use maximum value as an upper bound when $-5 \leq i \leq -2$, and use the maximum absolute sum as an upper bound when $-20 \leq i \leq -6$.

In[32]:= $\text{coeff} = \text{Table}[\text{If}[i \geq -5,$
 $\quad \text{Maximize}[\{\text{Abs}[\text{Coefficient}[G5, x^i]],$
 $\quad -1 \leq a \leq 3 \ \&\& -3994 \leq b \leq 64 \ \&\& -1620 \leq c \leq 3\}, \{x, a, b, c\}][[1]],$
 $\quad \text{FromCoefficientRules}[\text{Map}[\text{Abs}, \text{CoefficientRules}[\text{Coefficient}[G5, x^i],$
 $\quad \{x, a, b, c\}], \{2\}], \{x, a, b, c\}] /. a \rightarrow 3 /. b \rightarrow 3994 /. c \rightarrow 1620$
 $],$
 $\{i,$
 $\quad -20,$
 $\quad -2\}]$

Out[32]= $\left\{545102954553600, 1801659993602400, \frac{5886126860798565}{2}, 3236151695060880,$
 $\frac{19524150631163025}{8}, \frac{23280405487750863}{16}, 741299023799055, 334643780679111,$
 $131234047150977, 44282400488163, 13051286303076, 3426985781691,$
 $\frac{26026504017233}{32}, \frac{639099553189}{4}, \frac{37583702399}{2}, 17107740, 3629457, 1222632, 3672\right\}$

The computer proof for Step 5(b).

In[33]:= $\text{Simplify}[\text{Total}[\text{coeff} \, x^{\text{Range}[-20, -2]}] < 1, x \geq 120]$

Out[33]= True

Step 6

Definition of G_6 .

In[34]:= $G6 = \text{Factor}[\mu (y - x) (mm - m0^2) /. mmsub];$

Definition of F_4 . Eq. (5.18)

In[35]:= **F4 =**

Collect[Factor[Denominator[Factor[PolynomialExtendedGCD[F3, G6, μ]]][[2]]][[1]] /
($x^2 (1+x)^2 (x-y) y^2 (1+y)$), y, Factor]

Out[35]= $x^5 (-4 + m0^2 + 4x + 4x^2) (-3 + m0^2 + 4x + 4x^2)$
 $(-3 + 3m0^2 + 6x + 3m0^2x + 21x^2 + m0^2x^2 + 16x^3 + 4x^4) -$
 $x^2 (3m0^2 - 6m0^4 + 3m0^6 + 12x - 51m0^2x + 36m0^4x + 3m0^6x - 204x^2 + 204m0^2x^2 +$
 $37m0^4x^2 + 5m0^6x^2 + 204x^3 + 305m0^2x^3 + 60m0^4x^3 + 3m0^6x^3 + 1048x^4 +$
 $179m0^2x^4 + 80m0^4x^4 + m0^6x^4 + 688x^5 + 352m0^2x^5 + 48m0^4x^5 + 252x^6 +$
 $448m0^2x^6 + 12m0^4x^6 + 704x^7 + 240m0^2x^7 + 832x^8 + 48m0^2x^8 + 384x^9 + 64x^{10}) y +$
 $x (12m0^2 - 16m0^4 + 4m0^6 + 48x - 110m0^2x + 21m0^4x + 3m0^6x - 244x^2 -$
 $56m0^2x^2 + 34m0^4x^2 + 2m0^6x^2 - 448x^3 + 38m0^2x^3 + 30m0^4x^3 -$
 $180x^4 + 8m0^2x^4 + 12m0^4x^4 - 248x^5 - 448x^6 - 288x^7 - 64x^8) y^2 +$
 $(-12m0^2 + 7m0^4 - m0^6 - 48x + 20m0^2x - 44x^2 + 32m0^2x^2 - 8x^3 +$
 $24m0^2x^3 - 44x^4 + 12m0^2x^4 - 48x^5 - 16x^6) y^3$

Step 7

Definition of $y^{(1)}$, $y^{(2)}$, $y^{(3)}$.

In[36]:= **y1 = -** $\left(2x^3 + 3x^2 + \frac{3}{2}m0(m0+1)x + \frac{3}{4}m0(m0+1) \right)$;
y2 = - $\left(2x^3 + 3x^2 + \frac{3}{2}m0(m0-1)x + \frac{3}{4}m0(m0-1) \right)$;
y3 = x;

The computer proof for Step 7(a).

In[39]:= **Simplify[(F4 /. y → y1) == 0, m0 == 0]**

Out[39]= True

The computer proof for Step 7(b).

In[40]:= **Simplify[(F4 /. y → y1 - 1/2) > 0, x ≥ 90 && 1 ≤ m0 ≤ 2]**

Out[40]= True

The computer proof for Step 7(c).

In[41]:= **Simplify[(F4 /. y → y1 + 1/2) < 0, x ≥ 90 && 1 ≤ m0 ≤ 2]**

Out[41]= True

The computer proof for Step 7(d).

In[42]:= **Simplify[(F4 /. y → y2 + 1/x) < 0, x ≥ 90 && m0 == 0]**

Out[42]= True

The computer proof for Step 7(e).

In[43]:= **Simplify[(F4 /. y → y2 - 1/2) < 0, x ≥ 90 && 1 ≤ m0 ≤ 2]**

Out[43]= True

The computer proof for Step 7(f).

```
In[44]:= Simplify[(F4 /. y → y^2 + 1/2) > 0, x ≥ 90 && 0 ≤ m0 ≤ 2]
```

```
Out[44]= True
```

The computer proof for Step 7(g).

```
In[45]:= Simplify[(F4 /. y → y^3 - 1) > 0, x ≥ 1 && 0 ≤ m0 ≤ 2]
```

```
Out[45]= True
```

The computer proof for Step 7(h).

```
In[46]:= Simplify[(F4 /. y → y^3 + 1) < 0, x ≥ 1 && 0 ≤ m0 ≤ 2]
```

```
Out[46]= True
```

Step 8

Speed up Step 8 using multiple kernels.

```
In[47]:= LaunchKernels[8];
```

The computer proof for Step 8.

```
In[48]:= Select[DeleteDuplicates[Flatten[ParallelTable[
  If[(μ /. #) > 0 && y0 ≤ -1 && y0 ≠ -(2 x0^3 + 3 x0^2), {x0, y0, μ /. #}, {}] & /@
  Solve[(F3 /. x → x0 /. y → y0) == 0, μ, Integers]
  , {x0, 1, 120}, {y0, -(2 x0^3 + 3 x0^2 + 3 x0 + 2), -(2 x0^3 + 3 x0^2 - 3 x0 - 3)}], 2]],
  Length[#] > 0 &]
```

```
Out[48]= {{1, -1, 1}}
```